Lawrence Berkeley Laboratory
UNIVERSITY OF CALIFORNIA

EARTH SCIENCES DIVISION

UCB-NE-108 User's Manual

C.H. Kang and W.W.-L. Lee

April 1989



Prepared for the U.S. Department of Energy under Contract Number DE-AC03-76SF00098

DISCLAIMER

LBL-27044
UCB-NE-4143
UC-70

UCB-NE-108

User's Manual

C. H. Kang and William W.-L. Lee

Department of Nuclear Engineering
University of California

and

Earth Sciences Division
Lawrence Berkeley Laboratory
1 Cyclotron Road
Berkeley, CA 94720

April 1989

Work supported by the Director, Office of Civilian Radioactive Waste Management, Office of System Integration and Regulation, Licensing and Compliance Division, of the U. S. Department of Energy under contract DE-AC03-76SF00098.

INTRODUCTION

The purpose of this manual is to provide users of UCB-NE-108 with the information necessary to use UCB-NE-108 effectively.

DESCRIPTION OF THE CODE

UCB-NE-108[1] is a computer code for calculating the fractional release rate[2] of readily soluble radionuclides that are released from nuclear waste emplaced in water-saturated porous media, and transported through layers of porous media. Waste placed in such environments will gradually dissolve. For many species such as actinides and rare earths, the process of dissolution is governed by the exterior flow field, and the chemical reaction rate or leaching rate.[3] In a spent-fuel waste package the soluble cesium and iodine accumulated in fuel-cladding gaps, voids, and grain boundaries of spent fuel rods are expected to dissolve rapidly when groundwater penetrates the fuel cladding.[4] UCB-NE-108 is a code for calculating the release rate at the interface of two layers of porous material, such as the backfill around a high-level waste package and natural rock, to check compliance with the U. S. Nuclear Regulatory Commission's (USNRC) subsystem performance objective. It is an implementation of the analytic solution given below.

THEORY

The situation studied is shown in Figure 1. Waste face an internal gap or void space and is exposed in turn to two layers of porous media. The first layer of thickness b can be a backfill in a nuclear waste package, and the second the natural rock.

We define the origin at the interface of the two layers.
The governing equations are

$$\frac{\partial c_1}{\partial t} = D_1 \frac{\partial^2 c_1}{\partial x^2} - \lambda c_1, \qquad -b < x < 0, t > 0, D_1 \equiv \frac{D}{K_1} \tag{1}$$

$$\frac{\partial c_2}{\partial t} = D_2 \frac{\partial^2 c_2}{\partial x^2} - \lambda c_2, \qquad x > 0, t > 0, D_2 \equiv \frac{D}{K_2} \tag{2}$$

where c is the species concentration in ground water [M/L^3],
D is the species diffusion coefficient [L^2/t],
K is the species retardation coefficient,
ϵ is the porosity,
λ is the decay constant [t^{-1}],
and subscripts *1* and *2* refer to the layer closer to the waste and further away respectively.
The side conditions are

$$c_1(x,0) = 0, \qquad -b < x < 0, \tag{3}$$

$$c_2(x,0) = 0, \qquad x > 0 \tag{4}$$

$$c_1(0,t) = c_2(0,t), \qquad t \geq 0 \tag{5}$$

$$-\epsilon_1 D \frac{\partial c_1}{\partial x} = -\epsilon_2 D \frac{\partial c_2}{\partial x} \quad \text{at} \quad x = 0, \qquad t \geq 0 \tag{6}$$

$$c_2(\infty,t) = 0, \qquad t \geq 0 \tag{7}$$

The species concentration in the two regions is coupled by a mass balance equation over the void volume V with surface area S

$$-V \frac{\partial c_1(-b,t)}{\partial t} = -D\epsilon_1 S \frac{\partial c_1(-b,t)}{\partial x} + \lambda V c_1(-b,t), \qquad t \geq 0 \tag{8}$$

The initial concentration of the readily soluble species is taken to be

$$c_1(-b,0) = c^o \qquad \text{as specified} \tag{9}$$

The mass transfer rate at the backfill/rock interface is

$$\dot{M}(0,t) = 2K_1\epsilon_1 c^o S \frac{e^{-\lambda t}}{(\delta+1)} \sum_{n=0}^{\infty} \left[\sqrt{\frac{D_1}{\pi t}} \exp\left\{ \frac{(-2n+1)^2 b^2}{4D_1 t} \right\} - \gamma D_1 \exp\left\{ (2n+1)b\gamma + D_1\gamma^2 t \right\} \right.$$

$$\left. \times \text{erfc}\left\{ \frac{(2n+1)b}{2\sqrt{D_1 t}} + \gamma\sqrt{D_1 t} \right\} \left(\frac{\delta-1}{\delta+1} \right)^n \right], \qquad t \geq 0 \tag{10}$$

where

$$\delta = \sqrt{\frac{K_1}{K_2}} \frac{\epsilon_1}{\epsilon_2} \qquad \gamma = K_1\epsilon_1 S/V \tag{11}$$

The fractional release rate is obtained by dividing the mass flux by the 1000-year inventory of the species.

Unlike UCB-NE-107,[5] UCB-NE-108 does *not* include a contribution from the waste matrix.

ALGORITHM

UCB-NE-108 is an implementation of (11). UCB-NE-108 computes the mass flux as well as the fractional release rate.

The evaluation of

$$\text{F}(z^2) \equiv e^{z^2}\text{erfc}\sqrt{z}$$

is by an adaptation of the rational approximation of the error function.[6]

Numerical Illustration

Figure 2 shows some example fractional release rates of ^{135}Cs,^{137}Cs and ^{129}I based on (11). In this illustration we consider the release from a bare waste form exposed to ground water shortly after emplacement. We assume that fuel cladding and a container are not present, water contacts the interior of spent-fuel rods shortly after emplacement, and 1 percent of the total inventory of cesium and iodine is rapidly dissolved into the "void water" that fills voids in the waste package. The void water is equivalent in volume to a 7.4-cm thick layer of water between the waste solid and backfill. Ground-water flow is assumed to be small enough that mass transfer through backfill and into the rock is controlled by molecular diffusion. Time-dependent fractional release rates at the backfill/rock interface, normalized to initial inventories, are shown in Figure 2 for a diffusion coefficient of 10^{-5} cm^2/s, backfill porosity of 0.2, rock porosity of 0.01, a concentration-based distribution coefficient of 100 for cesium, and for a backfill thickness of 30 cm. Nonsorbing iodine-129 arrives at the backfill/rock interface in less than a year, with a peak release rate about tenfold less than the equivalent fractional release rate limit calculated from the NRC criterion. Cesium-135 and cesium-137 arrive later simultaneously, but the normalized peak release rate of cesium-137 is less because of more rapid decay. The peak release rate of cesium-135 is about tenfold less than its release rate limit, but the peak release rate of cesium-137 exceeds its limit by several orders of magnitude for several hundred years.

INPUT

The input consists of 3 lines. The input can be directed to a file by a command such as

ASSIGN inputfile FOR005

in the DEC/VAX VMS Operating System.

The input lines are
First Line (in free format)
- n0 = isotope initial inventory in the gap, g/m^3 [c° in (9)]
- s = surface area of the gap, m^2
- v = volume of the gap, m^3
- l = backfill thickness, m
- df = diffusion coefficient, m^2/a
- lambda = decay constant (if < 1, 1/a, or half-life, if > 1, a)

Second Line (in free format)
- k1 = backfill retardation coefficient
- k2 = rock retardation coefficient
- eps1 = backfill porosity
- eps2 = rock porosity

Third Line (in free format)
- istep = maximum time to compute, a [tmax=9*10**(istep-4)]
- weight = isotope initial inventory, g
- iplot = generate a plotting file (if =1)
- init=0 if fractional release rate is to be normalized to initial inventory

OUTPUT

There are two output streams. The first is a tabular output, currently directed to the screen. The second is a stream of output data, for use by a X-Y plotter or plotting program such as TELL-A-GRAF, if (iplot=1).

The tabular output is currently directed to the screen. In a DEC/VAX computer this output can be directed to a file by a command such as

ASSIGN filename FOR006

The tabular output prints the time, the mass transfer rate and the fractional release rate at the backfill/rock interface, and the number of terms used in the infinite series computation.

The output for X-Y plotting consists of only time and the fractional release rate in free format.

RESOURCE REQUIREMENTS

UCB-NE-108 is written in FORTRAN and does not call any outside libraries. It operates on DEC/VAX/VMS and DEC/UNIX machines.

VERIFICATION & BENCHMARKING

UCB-NE-108 is an implementation of (11). It was verified through hand calculations. The correctness of UCB-NE-108 was also checked this way. Eq. (11) was given to several workers and they were asked to write computer programs independently. The outputs of the various programs are then compared or benchmarked. They gave identical results.

REFERENCES

1. P. L. Chambré, C. H. Kang, W. W.-L. Lee and T. H. Pigford, "Mass Transfer of Soluble Species into Backfill and Rock," *Trans. Am. Nuc. Soc., 53,* 136, 1986.

2. U. S. Nuclear Regulatory Commission, Disposal of High-Level Radioactive Wastes in Geologic Repositories, 10 *Code of Federal Regulations* 60.113(a)(1)(ii)(B), 1983.

3. P. L. Chambré, C. H. Kang, W. W.-L. Lee and T. H. Pigford, "The Role of Chemical Reaction in Waste-Form Performance," in M. J. Apted and R. E. Westerman (eds.), *Scientific Basis for Nuclear Waste Management XI,* 285, Pittsburgh, Materials Research Society, 1988.

4. L. H. Johnson and D. W. Shoesmith, "Spent Fuel," in W. Lutze and R. C. Ewing (eds.), *Radioactive Waste Forms for the Future,* New York: Elsevier, 635, 1988.

5. W. W.-L. Lee, *UCB-NE-107 User's Manual,* LBL-26672, UCB-NE-4132, 1989.

6. K. H. Haskell and R. E. Jones, 1978. *Brief Instructions for Using the Sandia Mathematical Subroutine Library (Version 7.2),* SAND77-1441.



Figure 1. Release of Soluble Species in Two Layers of Porous Media



Figure 2 Fractional Release Rates for Various Nuclides Through 30 cm of Backfill

```
      program UCB_NE_108
c-------------------------------------------------------
c
c     This program calculates the fractional release rate
c     into the rock normalized by the initial isotope
c     inventory in semi-infinite domain of planar geometry,
c     Waste Form-Gap-Backfill-Rock geometry.
c
c
c     Input Parameters
c
c       First Card (in free format)
c            n0 = isotope initial inventory in the gap, g/m**3
c             s = surface area of the gap, m**2
c             v = volume of the gap, m**3
c             l = backfill thickness
c            df = diffusion coefficient, m**2/yr
c        lambda = decay constant (if <1), 1/yr, or
c                 half-life (if >1), yr
c
c       Second Card (in free format)
c            k1 = backfill retardation coefficient
c            k2 = rock retardation coefficient
c          eps1 = backfill poroity
c          eps2 = rock porosty
c
c       Third Card (in free format)
c         istep = maximum time to compute
c                   tmax=9*10**(istep-4)
c        weight = isotope initial inventory
c         iplot = generate plotting file (if =1)
c         init=0 if fractional release rate is to be normalized
c                to initial inventory
c
c     OUTPUT
c
c       Time, mass transfer rate in the rock, fractional release rate
c       (normalized to 1000-year inventory) and
c       number of terms used in the computation of the infinite series.
c       If iplot=1, then  the output file FOR004.DAT is
c       generated, and it prints time and fractional
c       release rate for TELL-A-GRAF.
c
c                                          Chul-Hyung Kang
c
c-------------------------------------------------------------
      implicit double precision (a-h,o-z)
      double precision k1,k2,n0,mdot,l,lambda
      common/aaa/pi
      read(5,*)n0,s,v,l,df,lambda
      read(5,*)k1,k2,eps1,eps2
      read(5,*)istep,weight,iplot,init
      write(6,*)n0,s,v,l,df,lambda
      write(6,*)k1,k2,eps1,eps2
      write(6,*)istep,weight,iplot,init
      write(6,110)
  110 format(///'  time, flux, frac. rel. rate, no. of term'/)
      if(lambda.gt.1.)lambda=log(2.d0)/lambda
      d1=df/k1
      gamma=k1*eps1*s/v
      delta=sqrt(k1/k2)*(eps1/eps2)
      pi=acos(-1.)
      fact=k1*eps1*n0*s*2./(delta+1.)
      do 100 i=1,istep
      do 100 j=1,9
      t=j*10.**(i-4)
```

```
      decay=lambda*t
      if(decay.gt.85.)go to 100
      sum=0.
      n=0
   10 z1=(2.*n+1.)*l/(2.*sqrt(d1*t))
      if(z1*z1 .gt. 85.)go to 20
      z=z1+gamma*sqrt(d1*t)
      if(delta.eq.1. .and. n.eq.0)then
      fact1=exp(-z1*z1)
      else
      fact1=exp(-z1*z1)*((delta-1.)/(delta+1.))**n
      endif
      fact1=fact1*(sqrt(d1/pi/t)-gamma*d1*herfl(z))
      sum=sum+fact1
      if (sum.eq.0.)go to 1
      if(abs(fact1/sum) .lt. 1.d-5)go to 20
    1 n=n+1
      go to 10
   20 mdot=fact*sum*exp(-decay)
      if (init.eq.0) then
          frac=mdot/weight
      else
          frac=mdot/(weight*exp(-lambda*1000.))
      endif
      if(frac.ge.1.e-18)then
      write(6,200)t,mdot,frac,n
      if(iplot.eq.1)write(4,*)t,frac
      endif
  100 continue
  200 format(1pd15.5,2d15.5,i5)
      stop
      end
      FUNCTION HERFL(X)
      REAL*8 X,AX,HERFL,TMP,X2,A,B,SQPI
      REAL*8 P(6),Q(6),R(4),S(4),ONE
      DATA P/22.898992851659D0,26.094746956075D0,14.571898596926D0,
     +4.2677201070898D0,.56437160686381D0,-6.0858151959688D-6/,
     +Q/22.898985749891D0,51.933570687552D0,50.273202863803D0,
     +26.288795758761D0,7.5688482293618D0,1D0/,
     +R/-1.21308276389978D-2,-.119903955268146D0,
     +-.2439110294886626D0,-3.24319519277746D-2/,
     +S/4.3002664345277D-2,.489552441961437D0,
     +  1.43771227937118D0,1D0/
      DATA SQPI/.564189583547756D0/, ONE /1D0/
      AX=DABS(X)
      IF(AX.GT.4.0) GO TO 10
      TMP=P(1)+AX*(P(2)+AX*(P(3)+AX*(P(4)+AX*(P(5)+AX*P(6)))))
      A=TMP/(Q(1)+AX*(Q(2)+AX*(Q(3)
     +                      +AX*(Q(4)+AX*(Q(5)+AX*Q(6))))))
      HERFL=DSIGN(A,X)
      GO TO 100
   10 CONTINUE
      X2=ONE/AX/AX
      B=X2*(R(1)+X2*(R(2)+X2*(R(3)+X2*R(4))))
     +  /(S(1)+X2*(S(2)+X2*(S(3)+X2*S(4))))
      A=(SQPI+B)/AX
      HERFL=DSIGN(A,X)
  100 CONTINUE
      RETURN
      END
```

```
Csal>run ne108
  9.2700000000000000        6.0800000000000000        0.4500000000000000
 7.4000000000000000E-02  3.1500000000000000E-03  2.3100000000000000E-07
  100.0000000000000        2400.000000000000        0.2000000000000000
 1.0000000000000000E-02
         11    417.0000000000000                 0            0


 time, flux, frac. rel. rate, no. of term

   2.00000D+00    2.49715D-10    5.98837D-13    1
   3.00000D+00    2.47182D-07    5.92762D-10    1
   4.00000D+00    7.06819D-06    1.69501D-08    1
   5.00000D+00    4.97290D-05    1.19254D-07    1
   6.00000D+00    1.75012D-04    4.19693D-07    1
   7.00000D+00    4.16718D-04    9.99325D-07    1
   8.00000D+00    7.79883D-04    1.87022D-06    1
   9.00000D+00    1.24597D-03    2.98793D-06    1
   1.00000D+01    1.78492D-03    4.28039D-06    1
   2.00000D+01    6.59391D-03    1.58127D-05    1
   3.00000D+01    7.90905D-03    1.89665D-05    2
   4.00000D+01    7.64124D-03    1.83243D-05    2
   5.00000D+01    6.95035D-03    1.66675D-05    2
   6.00000D+01    6.22154D-03    1.49198D-05    2
   7.00000D+01    5.56802D-03    1.33526D-05    2
   8.00000D+01    5.01237D-03    1.20201D-05    2
   9.00000D+01    4.54810D-03    1.09067D-05    3
   1.00000D+02    4.16100D-03    9.97841D-06    3
   2.00000D+02    2.32621D-03    5.57845D-06    4
   3.00000D+02    1.67096D-03    4.00711D-06    4
   4.00000D+02    1.31827D-03    3.16132D-06    5
   5.00000D+02    1.09299D-03    2.62108D-06    6
   6.00000D+02    9.34941D-04    2.24206D-06    6
   7.00000D+02    8.17213D-04    1.95974D-06    6
   8.00000D+02    7.25778D-04    1.74047D-06    7
   9.00000D+02    6.52531D-04    1.56482D-06    7
   1.00000D+03    5.92440D-04    1.42072D-06    7
   2.00000D+03    3.02094D-04    7.24447D-07   10
   3.00000D+03    1.96995D-04    4.72410D-07   11
   4.00000D+03    1.43147D-04    3.43279D-07   13
   5.00000D+03    1.10722D-04    2.65521D-07   14
   6.00000D+03    8.92360D-05    2.13995D-07   14
   7.00000D+03    7.40572D-05    1.77595D-07   15
   8.00000D+03    6.28285D-05    1.50668D-07   16
   9.00000D+03    5.42269D-05    1.30040D-07   16
   1.00000D+04    4.74550D-05    1.13801D-07   17
   2.00000D+04    1.90241D-05    4.56213D-08   20
   3.00000D+04    1.08665D-05    2.60588D-08   21
   4.00000D+04    7.23646D-06    1.73536D-08   22
   5.00000D+04    5.25531D-06    1.26027D-08   23
   6.00000D+04    4.03584D-06    9.67826D-09   23
   7.00000D+04    3.22287D-06    7.72870D-09   24
   8.00000D+04    2.64907D-06    6.35270D-09   24
   9.00000D+04    2.22639D-06    5.33906D-09   24
   1.00000D+05    1.90446D-06    4.56706D-09   25
   2.00000D+05    6.70710D-07    1.60842D-09   26
   3.00000D+05    3.59092D-07    8.61131D-10   26
   4.00000D+05    2.28657D-07    5.48338D-10   26
   5.00000D+05    1.60188D-07    3.84145D-10   26
   6.00000D+05    1.19229D-07    2.85920D-10   26
   7.00000D+05    9.25369D-08    2.21911D-10   26
   8.00000D+05    7.40589D-08    1.77599D-10   26
   9.00000D+05    6.06780D-08    1.45511D-10   27
   1.00000D+06    5.06438D-08    1.21448D-10   27
   2.00000D+06    1.42303D-08    3.41253D-11   27
```

```
   3.00000D+06     6.14773D-09     1.47428D-11   27
   4.00000D+06     3.16808D-09     7.59731D-12   27
   5.00000D+06     1.79830D-09     4.31246D-12   27
   6.00000D+06     1.08515D-09     2.60227D-12   27
   7.00000D+06     6.83048D-10     1.63800D-12   27
   8.00000D+06     4.43440D-10     1.06341D-12   27
   9.00000D+06     2.94761D-10     7.06861D-13   27
   1.00000D+07     1.99615D-10     4.78693D-13   27
   2.00000D+07     6.95207D-12     1.66716D-14   27
   3.00000D+07     3.72697D-13     8.93758D-16   27
   4.00000D+07     2.38388D-14     5.71675D-17   27
   5.00000D+07     1.67967D-15     4.02798D-18   27
FORTRAN STOP
```

LAWRENCE BERKELEY LABORATORY
TECHNICAL INFORMATION DEPARTMENT
UNIVERSITY OF CALIFORNIA
BERKELEY, CALIFORNIA 94720